
APL Optimization Techniques 
For Commercial Applications 

Eugene Ying 
2013 

1 



Topics 
 

1. Table Look Up 
2. Arithmetic Optimization 
3. Scalar Extension 

Assumptions 

• Dyalog APL Version 13.2.17248.0 64 bit Classic 
• Using defaults that come with V13.2 
• Running on IBM power7 processor (3.1Ghz) 
• In AIX 6.1 

Topic #1 
Table Look Up 

Vector Membership Function 

Subject Argument s Principal Argument P 

18  65  87 … 20  37  65  81  98 … є 

0 1 0 … 

Ticker Symbols 
Customer Names 
User IDs 
Machine Names 
Model IDs 
Product Codes 
... 

Tables in  
Commercial Data Processing 
are not always numeric 

Commercial Applications 

Your Customers Fortune 500 List 

Exxon Mobil 
Chrysler Group 
Apple 
RadioShack 

Wal-Mart Stores 
Exxon Mobil 
Chevron 
Phillips 66 
Berkshire Hathaway 
Apple 
… 

Matrix 
   є 

Answer is 1 0 1 0 

Principal Argument 

Subject Argument 

Inner Product 
 

∨/subject∧.=⍉PRINCIPAL 

Matrix Membership Function 

∨/s∧.=⍉P 

∨/s∧.=⍉P requires a matrix transpose that 
takes up temporary WS and processing time, 
followed by an inner product that creates 
a sparse matrix that takes up more 
temporary WS. 
 

Temporary Memory Requirement 

using inner product 
∨/subject∧.=⍉PRINCIPAL 

Matrix Membership Function 

using nested vector 
(↓subject)є↓PRINCIPAL 

(↓s)є↓P 

sєP 
can be emulated by 
 (ρP)≥P⍳s 

Similarly 
(↓s)є↓P 

can be emulated by 
  (1↑ρP)≥(↓P)⍳↓s 

Relationship Between 
Membership є and Index Of ⍳   

    or 
(1↑ρP)≥P{(↓⍺)⍳↓⍵}s 

(↓s)є↓P 

(1↑ρP)≥P{(↓⍺)⍳↓⍵}s 

If PRINCIPAL argument P is inside 
a loop and remains constant, 

:For I :in ⍳LOOP  
    B←(↓s)є↓P 
:Endfor 

Z←↓P 
HASH←є∘Z 
:For I :in ⍳LOOP  
    B←HASH ↓s 
:Endfor 

then P can be taken outside 
the loop for hash table.  

If PRINCIPAL argument P is inside 
a loop and remains constant, 

:For I :in ⍳LOOP          
    B←(1↑ρP)≥P{(↓⍺)⍳↓⍵}s 
:Endfor 

HASH←P∘{(↓⍺)⍳↓⍵} 
:For I :in ⍳LOOP  
    B←(1↑ρP)≥HASH s 
:Endfor                                                                                                                                                                                                                                                                                                                                             

then P can be taken outside 
the loop for a hash table. 

Precaution 
 
The P{(↓⍺)⍳↓⍵}s Idiom 
is optimized for character matrix and 
integer matrix arguments.  
 
This idiom is many times less efficient if 
your matrix arguments have floating point 
numbers.  
 
In such cases you may wish to explore 
algorithms more efficient than this Idiom 
to substitute for the expression ∨/s∧.=⍉P 

Recommendation 
 
In your APL function libraries, whenever 
you see 
 
∨/s∧.=⍉P 
or 
(↓s)є↓P 
 
if the character matrices are large, 
replace them by 
 
(1↑ρP)≥P{(↓⍺)⍳↓⍵}s 
 
and you will see much faster execution 
speed with less workspace requirement.  

Recommendation 
 
Furthermore, if the Matrix Epsilon is inside 
a loop, and if the PRINCIPAL character 
matrix P is constant, then replace 
 
:For I :In ⍳LOOP 
    B←(1↑ρP)≥P{(↓⍺)⍳↓⍵}s 
:Endfor 
 
by 
 
HASH←P∘{(↓⍺)⍳↓⍵} 
:For I :In ⍳LOOP 
    B←(1↑ρP)≥HASH s 
:Endfor 

Another Situation 
 
Up to now, we are concerned with ∨/s∧.=⍉P 
but if you see the expression ∨⌿A∧.=⍉B 
you can do this optimization: 
∨⌿A∧.=⍉B 
   ↓ 
∨/B∧.=⍉A 
   ↓ 
(↓B)є↓A 
   ↓ 
(1↑ρA)≥(↓A)⍳↓B 
   ↓ 
(1↑ρA)≥A{(↓⍺)⍳↓⍵}B 
 
And see the previous page in case the hash 
table approach is applicable. 
 27 



Topic # 2 
Arithmetic Optimization 

M←5000 10ρ0.1+⍳50000 

:For I :In ⍳10000 
    {}M÷100 
:EndFor 

:For I :In ⍳10000 
    {}M×0.01   
:EndFor  

⍝ 24492 ms   

⍝   901 ms 

Divide vs Multiply 

29 



:For I :In ⍳10000 
    {}M÷7 
:EndFor 

R7←÷7 
:For I :In ⍳10000 
    {}M×R7  
:EndFor  

:For I :In ⍳10000 
    {}M×0.1428571429  
:EndFor  

Readability of Constants 

In APL, the Reciprocal function 
is usually faster than the Divide 
function 

:For I :In ⍳1000000 
    {}1÷7  
    {}÷7 
:EndFor  

⍝ 1234 ms 
⍝ 1051 ms 

Divide vs Reciprocal 

Since 1÷Array is slower than ÷Array 
should we change 0-Array to –Array?   

M←10000 10ρ0.1+⍳100000 
:For I :In ⍳1000000 
    {}0-M  
    {}-M 
:EndFor  

⍝ 110814 ms 
⍝   8546 ms 

An APL interpreter can sometimes make 
use of extra help from an APL 
programmer to make division and 
subtraction execute faster. 
 
But should the APL programmer make 
the function arguments conformable to 
further help the APL interpreter? 
 
e.g.   Should M×0.01 
be changed to M×(ρM)ρ0.01 

M←5000 10ρ0.1+⍳50000 

:For I :In ⍳10000 
    {}M÷100 
:EndFor 

:For I :In ⍳10000 
    {}M×0.01   
:EndFor  

⍝ 24570 ms   

⍝   653 ms 

:For I :In ⍳10000 
    {}M×(ρM)ρ0.01  
:EndFor  

⍝  1545 ms 

Topic #3. Scalar Extension 

M←5000 10ρ0.1+⍳50000 ⍝ Matrix M     
S←30     ⍝ Scalar S 

:For I :In ⍳10000 
    {}S×M×0.01   
:EndFor  

⍝ 1566 ms 

:For I :In ⍳10000 
    {}M×S×0.01  
:EndFor  

⍝  773 ms 

Order of Execution for Scalars 

Scalar1×Array×Scalar2  

Array×Scalar1×Scalar2  

can be optimized as 

so that two scalar extensions 
are reduced to only one scalar 
extension. 

Matrix є Scalar 
Matrix = Scalar 

:For I :In ⍳10000 
    {}Mє1000.1 
    {}M=1000.1 
:EndFor 

⍝ 10207 ms   
⍝  4385 ms 

Should we use ’є’ ? 
Or should we use ‘=‘ ?  

Because of scalar extension, 
M=1000.1 is faster than Mє1000.1 

M←5000 10ρ0.1+⍳50000 

      26⊥¯1+⎕A⍳'DYALOG' 
46619358 

Encoding & Decoding Characters 

      ⎕A[1+(6ρ26)⊤4619358] 
DYALOG 

The ¯1+ is performed 6 times 
because of scalar extension.  

The 1+ is performed 6 times 
because of scalar extension.  

In a ⎕IO=1 environment  

Avoid Scalar Extension 

      {⎕IO←0 ◊ ⎕A[(6ρ26)⊤⍵]}46619358 
DYALOG 

Now there is no need for 
scalar extension of ¯1+ 

Now there is no need for 
scalar extension of 1+ 

      {⎕IO←0 ◊ 26⊥⎕A⍳⍵}'DYALOG' 
46619358 

Advantage of Localizing  
System Variable in D-fn 

When an error occurs, 
thanks to D-fn,  
⎕IO bounces back to 1  

SYNTAX ERROR 
      {⎕IO←0 ◊ 26⊥⎕A⍳⍵}= 
     ^ 
      ⎕IO 
1 

Recommendation 
 
Scalar extension has been implemented 
very efficiently in APL. Whenever 
there is a chance, let APL do the 
scalar extension for you. Do not 
reshape the scalar to help the APL 
interpreter. 
 
If an expression has a combination of 
scalars and arrays, let APL work with 
scalars first, then do scalar 
extension later. 

When you enclose a numeric array to make 
it a scalar, this scalar array can take 
advantage of scalar extension to retrieve 
items from a nested array efficiently. 
 
This is best illustrated by examples. 

Scalar Array 

43 



  DISPLAY A 
.→---------------------------------------------------. 
↓ .→-------------. .→-------------. .→-------------. | 
| ↓1111 1112 1113| ↓1211 1212 1213| ↓1311 1312 1313| | 
| |1121 1122 1123| |1221 1222 1223| |1321 1322 1323| | 
| |1131 1132 1133| |1231 1232 1233| |1331 1332 1333| | 
| '~-------------' '~-------------' '~-------------' | 
| .→-------------. .→-------------. .→-------------. | 
| ↓2111 2112 2113| ↓2211 2212 2213| ↓2311 2312 2313| | 
| |2121 2122 2123| |2221 2222 2223| |2321 2322 2323| | 
| |2131 2132 2133| |2231 2232 2233| |2331 2332 2333| | 
| '~-------------' '~-------------' '~-------------' | 
'є---------------------------------------------------' 

A 2 by 3 Nested Matrix Where Each Item Is 
A 3 by 3 Numeric Matrix 

                              A[((1 1)(2 2)) ((2 3)(2 2))] 

.→---------------------------------------------------. 
↓ A[⊂((1 1)(2 2))]                                   | 
| .→-------------. .→-------------. .→-------------. | 
| ↓1111 1112 1113| ↓1211 1212 1213| ↓1311 1312 1313| | 
| |1121 1122 1123| |1221 1222 1223| |1321 1322 1323| | 
| |1131 1132 1133| |1231 1232 1233| |1331 1332 1333| | 
| '~-------------' '~-------------' '~-------------' | 
|                                                    | 
|                                   A[⊂((2 3)(2 2))] | 
| .→-------------. .→-------------. .→-------------. | 
| ↓2111 2112 2113| ↓2211 2212 2213| ↓2311 2312 2313| | 
| |2121 2122 2123| |2221 2222 2223| |2321 2322 2323| | 
| |2131 2132 2133| |2231 2232 2233| |2331 2332 2333| | 
| '~-------------' '~-------------' '~-------------' | 
'є---------------------------------------------------' 

To select the red items, you can use 
A[((1 1)(2 2)) ((2 3)(2 2))] 

                       (⊂2 2)⌷¨ 
.→---------------------------------------------------. 
↓       2 2⌷             2 2⌷             2 2⌷       | 
| .→-------------. .→-------------. .→-------------. | 
| ↓1111 1112 1113| ↓1211 1212 1213| ↓1311 1312 1313| | 
| |1121 1122 1123| |1221 1222 1223| |1321 1322 1323| | 
| |1131 1132 1133| |1231 1232 1233| |1331 1332 1333| | 
| '~-------------' '~-------------' '~-------------' | 
|                                                    | 
|       2 2⌷             2 2⌷             2 2⌷       | 
| .→-------------. .→-------------. .→-------------. | 
| ↓2111 2112 2113| ↓2211 2212 2213| ↓2311 2312 2313| | 
| |2121 2122 2123| |2221 2222 2223| |2321 2322 2323| | 
| |2131 2132 2133| |2231 2232 2233| |2331 2332 2333| | 
| '~-------------' '~-------------' '~-------------' | 
'є---------------------------------------------------' 

To select the red items, you can use 
A[2 3ρ((1 1)(2 2)) ((1 2)(2 2)) ((1 3)(2 2)) ((2 1)(2 2))     
((2 2)(2 2)) ((2 3)(2 2))] ⍝ 8.1 sec 1,000,000 loops 

or (⊂2 2)⌷¨A ⍝ Squad Indexing 6.7 sec 1,000,000 loops 

                                                          (⊂2 2)↑¨
.→---------------------------------------------------. 
↓       2 2↑             2 2↑             2 2↑       | 
| .→-------------. .→-------------. .→-------------. | 
| ↓1111 1112 1113| ↓1211 1212 1213| ↓1311 1312 1313| | 
| |1121 1122 1123| |1221 1222 1223| |1321 1322 1323| | 
| |1131 1132 1133| |1231 1232 1233| |1331 1332 1333| | 
| '~-------------' '~-------------' '~-------------' | 
|                                                    | 
|       2 2↑             2 2↑             2 2↑       | 
| .→-------------. .→-------------. .→-------------. | 
| ↓2111 2112 2113| ↓2211 2212 2213| ↓2311 2312 2313| | 
| |2121 2122 2123| |2221 2222 2223| |2321 2322 2323| | 
| |2131 2132 2133| |2231 2232 2233| |2331 2332 2333| | 
| '~-------------' '~-------------' '~-------------' | 
'є---------------------------------------------------' 

To select the red items, you can use 
2↑[2]¨2↑[1]¨A   ⍝ 13.3 sec (looping 1,000 times) 
or 
(⊂2 2)↑¨A       ⍝  4.9 sec (looping 1,000 times) 

                   (⊂(2 3)(,2))⌷¨A 
.→---------------------------------------------------. 
↓   (2 3)(,2)⌷       (2 3)(,2)⌷        (2 3)(,2)⌷    | 
| .→-------------. .→-------------. .→-------------. | 
| ↓1111 1112 1113| ↓1211 1212 1213| ↓1311 1312 1313| | 
| |1121 1122 1123| |1221 1222 1223| |1321 1322 1323| | 
| |1131 1132 1133| |1231 1232 1233| |1331 1332 1333| | 
| '~-------------' '~-------------' '~-------------' | 
|   (2 3)(,2)⌷       (2 3)(,2)⌷       (2 3)(,2)⌷     | 
| .→-------------. .→-------------. .→-------------. | 
| ↓2111 2112 2113| ↓2211 2212 2213| ↓2311 2312 2313| | 
| |2121 2122 2123| |2221 2222 2223| |2321 2322 2323| | 
| |2131 2132 2133| |2231 2232 2233| |2331 2332 2333| | 
| '~-------------' '~-------------' '~-------------' | 
'є---------------------------------------------------' 

To select the red items, you can use 
¯1↑[2]¨(⊂¯2 2)↑¨A  ⍝ 10 sec (looping 1,000 times) 
or 
(⊂(2 3)(,2))⌷¨A  ⍝  7 sec (looping 1,000 times) 

                     (⊂⊂1 3)⌷[1]¨ 
.→---------------------------------------------------. 
↓    (⊂1 3)⌷[1]       (⊂1 3)⌷[1]       (⊂1 3)⌷[1]    | 
| .→-------------. .→-------------. .→-------------. | 
| ↓1111 1112 1113| ↓1211 1212 1213| ↓1311 1312 1313| | 
| |1121 1122 1123| |1221 1222 1223| |1321 1322 1323| | 
| |1131 1132 1133| |1231 1232 1233| |1331 1332 1333| | 
| '~-------------' '~-------------' '~-------------' | 
|                                                    | 
|    (⊂1 3)⌷[1]       (⊂1 3)⌷[1]       (⊂1 3)⌷[1]    | 
| .→-------------. .→-------------. .→-------------. | 
| ↓2111 2112 2113| ↓2211 2212 2213| ↓2311 2312 2313| | 
| |2121 2122 2123| |2221 2222 2223| |2321 2322 2323| | 
| |2131 2132 2133| |2231 2232 2233| |2331 2332 2333| | 
| '~-------------' '~-------------' '~-------------' | 
'є---------------------------------------------------' 

To select the red items, you can use 
(1↑[1]¨A),[1]¨¯1↑[1]¨A     17.4 sec (looping 1,000 times) 
or 
(⊂⊂1 3)⌷[1]¨A    7.5 sec (looping 1,000 times) 

Recommendation 
 
When you need to uniformly index a 
matrix consisting of nested vectors 
as items, squad indexing with the 
left argument enclosed as a scalar 
can be a good alternative.  
 
The scalar extension could simplify 
the expression and speed up the 
execution time. 

APL Optimization Techniques 
For Commercial Applications 

Questions? 
 
 

Eugene Ying 
2013 

